

max. file "MACSYMA"

MACSYMA Primer - Introductory Section

MACSYMA (pronounced "maxima"), Project MAC's SYmbolic MAManipulation system, is a large computer program devoted to the manipulation of algebraic expressions. The system provides many capabilities, some of which we shall endeavor to explain here. A complete list of the current commands is given in the MACSYMA manual. The first thing a new user must learn is the interaction with the time-sharing system. The time sharing system which MACSYMA uses is called ITS, and was originally developed at the Artificial Intelligence Laboratory at MIT.

Login procedure

If someone has previously used your terminal, you will see a message like
ITS 6 CONSOLE 17 FREE

In any case, in order to get ITS to listen to you, type

control Z

That is, hit the Z key while holding the control key down.

This will result in a message starting with

DDT 234.

Other material will probably be printed also which you should read.

At this point the system is waiting for you to log in. The log-in command is

:LOGIN JDOE ↵

We use ↵ to indicate a carriage return. The name JDOE is used for identification. Presumably you will use your own name or initials. Most names or initials are acceptable, as long as no one of that name is already logged in. In such a case, use a different name or spelling. In no case should you log in as MACSYMA or MACSYM.

Once you have logged in, you can use MACSYMA by typing

:MACSYMA ↵

After a few seconds MACSYMA will type back

THIS IS MACSYMA 150

(C1)

The number 150 is a version identification number. This number will vary from time to time. The label (C1) is a name being automatically assigned to your first command.

Suppose you wanted to work with the expression $(x+1)^3$, you could type it in by using FORTRAN-like syntax for expressions as follows:

(C1) (X + 1)**3@

The @ terminates your input string of characters and prompts MACSYMA to evaluate your expression, simplify it, and print out the result. In this case evaluation and simplification are null operations. MACSYMA will come back with

(D1) (X + 1)³

Note that your expression is displayed in a two-dimensional notation comparable to that of a text-book. Your result is also assigned a label D1 which may be used in subsequent commands. MACSYMA also automatically labelled the next input line C2.

Let us indicate one of the over one hundred commands available in MACSYMA, a command for expanding expressions. Commands are written in functional notation, to wit:

(C2) EXPAND(D1)@

(D2) $X^3 + 3 X^2 + 3 X + 1$

One of the first things you will want to learn is how to correct your input line. There are several possible methods.

For example, suppose you typed

(C2) EXPAM

Now you may delete the last character "M" by hitting the rub-out or delete key once. Doing this will cause the character to be repeated.

(C2) EXPAMM

Now you can add the remaining characters to your input string (i.e. `ND(D1)`):

(C2) `EXPANND(D1)`

(D2) $x^3 + 3x^2 + 3x + 1$

This use of the rub-out key will also help you in correcting mistakes in logging in. Sometimes you just wish to start all over again. To do this type `??`. For example,

(C2) `EXPA??`

(C2)

If you made a mistake in the syntax, but hit the `␣` before correcting it, you will encounter a message like

INPUT ERROR AT POINT MARKED BY `**␣**`

`E X P A N D ((D 1) **␣**`

PLEASE REPHRASE OR EDIT

Editing is a more complex facility than you will need at first. Hence we will not enter into it at this point, so you might as well retype your command taking care to avoid the error just made.

Let us consider a few additional commands and facilities. To differentiate an expression use `DIFF(expr, var)`. Here `expr` is the expression or its name, `var` is the variable with respect to which differentiation is to be performed.

(C3) `SIN(X)*COS(X)`

(D3) $\cos(x) \sin(x)$

(C4) `DIFF(%,X)`

(D4) $\cos^2(x) - \sin^2(x)$

Note the use of `%` in C4. The symbol `%` always represents the previous expression, in this case D3. To differentiate an expression twice use `DIFF(expr, var, 2)`.

(C5) `DIFF(D3,X,2)`

(D5) $-4\cos(x) \sin(x)$

There are a number of ways for effecting a substitution of one expression for another inside of a third. For example to substitute X^2 for every occurrence of Z in the expression D6 which is Ze^Z you could write

(C7) D6, Z = X**2@

(D7) $X^2 e^{X^2}$

Note that the base of natural logarithms is written %E, not E. The square root of -1 is written as %I and π is written %PI.

An alternative syntax for the command C7 is

(C8) EV(D6, Z = X**2)@

(D8) $X^2 e^{X^2}$

An equivalent command is

(C9) SUBSTITUTE (X**2,Z, D6)@

(D9) $X^2 e^{X^2}$

Note the order of arguments to SUBSTITUTE: substitute the first for every occurrence of the second inside the third.

We shall now consider some more linguistic facilities available in MACSYMA.

To assign an expression to a variable use:

(C10) A : % @

(D10) $X^2 e^{X^2}$

Note the variable A will have the value $X^2 e^{X^2}$. Thus

(C11) A + 1 @

(D11) $X^2 e^{X^2} + 1$

Frequent mistakes are due to forgotten assignments. To unassign a variable A say

(C12) KILL(A) @

(D12) TRUE

Now if you were to ask for the value of A, you'd get

(C13) A @

(D13) A

To define a function F(Z) to be $\sin^2(Z) + 1$ use :=

(C14) F(Z) := SIN(Z)**2 + 1 @

(D14) $F(Z) := \sin^2(Z) + 1$

(C15) F(X + 1)@

(D15) $\text{SIN}^2(X + 1) + 1$

Equations in MACSYMA are a particularly useful form of expressions. To represent the equation $X^2 + 2X = Y^2$, use

(C16) X**2+2*X=Y**2@

(D16) $X^2 + 2X = Y^2$

One may add expressions to equations, multiply an equation by an expression and add two equations together.

(C17) D16 + 1 @

(D17) $X^2 + 2X + 1 = Y^2 + 1$

The left-hand-side of an equation is obtainable by the function LHS. RHS obtains the right-hand-side.

(C18) LHS(%)@

(D18) $X^2 + 2X + 1$

The left-hand-side of an equation can also be obtained by the more general command PART. The left-hand-side is the first part of an equation, the right-hand-side is the second part.

(C19) PART(D17,2)@

(D19) $Y^2 + 1$

The first term of a sum is its first part.

(C20) PART(% ,1)@

(D20) Y^2

One would also have gotten this result from PART(D17,2,1).

Equations are generated as intermediate results of MACSYMA's SOLVE command.

(C21) X**2-1 @

(D21) $X^2 - 1$

(C22) SOLVE(%,X)@

SOLUTION

(E22) X = 1

(E23) X= -1

(D23) [E22, E23]

The final result of SOLVE is a list of its intermediate solutions. SOLVE can, among other things, obtain closed form solutions to polynomials which can be factored into linear, quadratic, cubic or quartics over the integers.

To substitute one of the solutions into the original equation you can write

(C24) D21, E22 @

(D24) 0

Since E22 evaluates to the equation $X=1$, the substitution is made into X^2-1 and the result is simplified.

The sum of the numbers i^2 , $i = 1, 2, \dots, 5$ can be obtained as follows:

(C25) SUM(I**2, I, 1, 5)@

(D25) 55

One could also obtain this result by using a FOR statement.

(C26) FOR I : 1 STEP 1 THRU 5 DO (S: S + I**2) @

(D26) DONE

The result is stored in the value of the variable S. Had S been initialized to 0, the result would be 55. Had S not been given a value then

(C27) S @

(D27) S + 55

This result appears curious, but is quite consistent with MACSYMA's evaluation mechanism. The value of an unassigned variable is itself. Thus the first step through the FOR statement we execute $S : S + 1$. The result is $S + 1$ and this

expression stored in S. To recover from the situation in D27, we can set S to 0 as in

(C28) %, S = 0 @

(D28) 55

To get an even more curious result, but one which has uses in many cases, we can reevaluate the expression causing all variables in it to be evaluated and function calls to be reexecuted.

(C29) D27, EVAL @

(D29) S + 110

Returning to summations, we sometimes want an expression containing a sum which is unevaluated or unevaluatable. For instance,

(C30) 'SUM(G(I), I, 0, N) @

(D30)
$$\sum_{I=0}^N G(I)$$

Note the use of an undefined function G which is displayed with its argument evaluated. G may be given a definition or substituted for at a later time. Also note the use of the quote symbol. The effect here is to prevent an attempt to evaluate the sum. In this case, however, the quote makes little difference since we would have obtained the same result had we not quoted because the upper limit, N, has not been assigned a value.

MACSYMA considers the quoted and unquoted form of a function to represent its noun and verb forms, respectively. Most functions are verbs and will be evaluated. The trigonometric functions (e.g. SIN, COS) are nouns and normally do not evaluate, even if given numerical arguments. Thus

(C31) SIN (1) @

(D31) SIN (1)

To evaluate trigonometric functions with numeric arguments use a NUMER specification.

(C32) SIN (1), NUMER @

(D32) 0.84147098

In general, a noun form of a function for which a verb form exists may be evaluated by giving the name of the function as an argument at the top level. Thus D7, G @ will evaluate instances of the function G in D7.

Let us return to a more general treatment of expansion. EXPAND uses the variables MAXPOSEX and MAXNEGEX to control the maximum positive and negative integer exponent which will be used in expansion. These values are normally 6 each. Thus

(C33) (X + 1)**7 @

(D33) (X + 1)⁷

(C34) EXPAND(%) @

(D34) (X + 1)⁷

To force expansions with higher values of MAXPOSEX and MAXNEGEX, you may set their values using : or locally as in EXPAND(%), MAXPOSEX = 20 @, or as arguments to EXPAND. Thus

(C35) EXPAND(D33, 7, 7)@

(D35) $X^7 + 7 X^6 + 21 X^5 + 35 X^4 + 35 X^3 + 21 X^2 + 7 X + 1$

In some situations you may wish to keep all expressions from a given point on in expanded form, rather than just expanding the final result. To do this set the values of EXPOP and EXPON appropriately. For example setting EXPOP to 1 expands through sums, but not through powers of sums.

(C36) EXPOP : 1 @

(D36) 1

(C37) A*(B+C) + A*(B+C)**2 @

(D37) A (B+C)² + A C + A B

Another facility in MACSYMA is its ability to integrate.

(C38) X/(X**3 + 1) @

(D38) $\frac{X}{X^3 + 1}$

(C39) INTEGRATE (% ,X) @

(D39) $\frac{\text{ATAN}\left(\frac{2X - 1}{\text{SQRT}(3)}\right)}{\text{SQRT}(3)} + \frac{\text{LOG}(X^2 - X + 1)}{6} - \frac{\text{LOG}(X + 1)}{3}$

To check the solution, try differentiation.

(C40) DIFF (% ,X) @

(D40) $\frac{2}{3\left(\frac{(2X - 1)^2}{3} + 1\right)} + \frac{2X - 1}{6(X^2 - X + 1)} - \frac{1}{3(X + 1)}$

Note that the result of differentiation is not quite the same as the original expression D38. The reason is that the terms are not expanded, nor is the sum put over a common demonimator. A quick fix is provided by the RAtional SIMPLification command RATSIMP.

(C41) RATSIMP (%) @

(D41) $\frac{X}{X^3 + 1}$

There are several things worth noting at this point. The RATSIMP command uses an entirely different way of representing expressions internally than most of the commands we have encountered so far (SOLVE is an exception). The advantage of a multiplicity of representations is that in some cases one form is much better than another. The variety of representations is quite handy in many situations, as most users will quickly recognize. The representation of expressions which RATSIMP

employs is that of rational functions in a canonical form which is a ratio of polynomials. By canonicalizing expression D40 one obtains D41. While D41 is smaller than D40, RATSIMP will not always return smaller results. In particular RATSIMP applied to $(X + 1)^{10}$ will return the expanded result, which is the canonical form of the polynomial. Sometimes such a transformation is useful, sometimes it is not. The user, of course, has the option of employing whichever form is suitable in his problem.

Many of MACSYMA's commands benefit from the rational function manipulation routines. One such command is FACTOR.

(C42) $X^{12} + 1$ @

(D42) $X^{12} + 1$

(C43) FACTOR (%) @

(D43) $(X^4 + 1)(X^8 - X^4 + 1)$

When SOLVE is given a polynomial it usually will call FACTOR to factor it. Should the factors be of degree ≤ 4 , then the standard formulas are applied to generate the solution. FACTOR is limited to returning results which have integer coefficients. Therefore it does not factor $X^8 - X^4 + 1$ which is a polynomial irreducible over the integers. This is an important restriction and has side-effects in many commands. For example, INTEGRATE will not be able to integrate $1/(X^8 - X^4 + 1)$.

At this point it should be clear that an algebraic manipulation system like MACSYMA cannot solve all symbolic problems. In addition to theoretical difficulties which cause SOLVE and INTEGRATE not to be able to find closed form solutions, one has practical problems. The most common problem is given the name "intermediate expression swell." It frequently happens that expressions in the middle of a calculation grow very large, even though the final result is relatively small. This may cause your calculation to run out of storage. With some experience you will learn how to control the size of your intermediate expressions. For the time being, you should note that a liberal use of the KILL command will generate enough storage for most reasonable computations.

Before continuing with more advanced features of MACSYMA, you should learn how to log out of the system.

The logout procedure from MACSYMA is similar to the login procedure. First type control Z.

as before by hitting the Z key while holding the control key down.

This will result in some not very meaningful message, but the effect is to get you back to the time-sharing system. Now type

```
:LOGOUT ↵
```

and you will get a message like

```
ITS 6 CONSOLE 17 FREE.
```

You will, from time to time, encounter bugs or features that you do not like or understand. To report problems in using MACSYMA, you can interrupt MACSYMA by typing a "control Z" and sending a message as follows

```
:SEND ␣ MACSYMA ␣ (MAIL) ... message ... control C
```

Here we use the mark ␣ to indicate a space.

After you type "SEND ␣ MACSYMA ␣", the system will type back (MAIL) indicating that the message will be left on the disk. Now type your message, make it as detailed as you

can, and use carriage returns liberally. You end your message with a control C. At this point you are ready to return to MACSYMA. To accomplish the return, type

alt P

That is, first hit the alt mode or escape key and then a "P". To read the latest messages to MACSYMA and some of our answers, use a similar procedure to the :SEND.

Instead of :SEND type

:PRMAIL MACSYMA

You can stop the printing at any convenient point by typing "control G" which is used throughout as the quit character. The return to MACSYMA is accomplished as before.

If you wish to know who is using the system at any point, get out of MACSYMA by typing a control Z and then type

TTY control F

A heavy load on the system will be indicated by having a large number of blocks (block = 1024 words) OUT on the disk. You should get into the habit of looking at the information supplied by TTY control F to determine if you would be better served by logging in at a later time.